UNIVERSITY OF ILLINOIS
URBANA

AERONOMY REPORT NO. 37

FORTRAN PROGRAMS FOR CALCULATING LOWER IONOSPHERE ELECTRON DENSITIES AND COLLISION FREQUENCIES FROM ROCKET DATA

by

E. A. Mechtly
P. E. Monro
N. Golshan
R. S. Sastry

July 1, 1970

Supported by
National Aeronautics and Space Administration
Grant ~~NGR 013~~ NGR-14-005-013

Aeronomy Laboratory
Department of Electrical Engineering
University of Illinois
Urbana, Illinois

CITATION POLICY

A E R O N O M Y   R E P O R T

N O. 3 7

FORTRAN PROGRAMS FOR CALCULATING LOWER IONOSPHERE ELECTRON DENSITIES AND COLLISION FREQUENCIES FROM ROCKET DATA

by

E. A. Mechtly
P. E. Monro
N. Golshan
R. S. Sastry

July 1, 1970

Supported by
National Aeronautics and
Space Administration
Grant ~~NGR-013~~ NGR-14-005-013

Aeronomy Laboratory
Department of Electrical Engineering
University of Illinois
Urbana, Illinois

UILU-ENG-70-260

ABSTRACT

This report is a record and discussion of computer programs used by personnel of the Aeronomy Laboratory for calculating D- and E-region electron densities and collision frequencies from Faraday rotation, differential absorption, and standing-wave data obtained from Nike Apache rocket-radio propagation experiments. All the programs are written in FORTRAN IV and are compatable with the IBM 360 system operated by the Department of Computer Science at the University of Illinois in Urbana.

TABLE OF CONTENTS

LIST OF FIGURES

1. INTRODUCTION

The rocket-radio propagation experiment of the University of Illinois was first described by Bowhill (1965). Instrumentation of the experiment is discussed by Knoebel and Skaperdas (1966). Fundamentals of the method of analysis of Faraday rotation and differential absorption data are discussed by Mechtly _et al_. (1967), and of standing waves, by Monro _et al_. (1968).

The present report records and describes computer programs which execute the detailed numerical analysis of Faraday rotation, differential absorption, and standing-wave data from the rocket experiments. All of these programs are expressed in the language of FORTRAN IV.

2. SIMULTANEOUS ANALYSIS OF FARADAY AND ABSORPTION DATA

Ideally, both Faraday rotation (FR) and differential absorption (DA) data are measured at all altitudes of interest in the lower ionosphere. Unfortunately, in the region between about 55 and 85 km, only absorption data are generally measurable. Here, the electron concentrations are too small to cause a measurable Faraday rotation. However, both types of data are usually obtained between about 85 and 100 km.

We first consider, in this section, a computer program to analyze both types of data simultaneously. In Section 3, we consider the analysis of absorption data in the absence of Faraday data.

A program for the simultaneous analysis of FR and DA data is listed in Tables 1, 2, and 3. This program has no branches except to subroutines. Therefore, the program will be described sequentially without reference to a flow chart.

Table 1 lists alphabetically the names, definitions and units of the principal variables used in the program. Tables 2 and 3 are the FORTRAN IV statements of the main part of the program. The subroutines are discussed in Sections 4 and 5.

The values of physical constants are those listed by Taylor et al. (1969), and Mechtly (1969).

References to table numbers and statement numbers are made by enclosing the table number and the statement number in parentheses; the two numbers separated by a dash. For example, the notation (2-2) is a reference to Statement No. 2 of Table No. 2.

Statements (2-2) to (2-9) enter the equatorial radius and flattening factor of an oblate earth and generate constants from them for later use.

TABLE 1

Names of Variables

A = equatorial radius of the earth (meters)
AIO = absorption index, ordinary mode
AIX = absorption index, extraordinary mode
AZD = azimuth angle of the rocket from the transmitter (deg)
B = the total flux density of the geomagnetic field (tesla)
BD = the downward component of the geomagnetic flux density (tesla)
BN = the north component of the geomagnetic flux density (tesla)
BP = the phi component of the geomagnetic flux density (tesla)
BR = the radial component of geomagnetic flux density (tesla)
BT = the theta component of the geomagnetic flux density (tesla)
BW = the westward component of the geomagnetic flux density (tesla)
CF = electron collision frequency, most probable (s^{-1})
CPH = cosine of the geocentric longitude angle of the rocket
CT = cosine of the geocentric polar angle of the rocket
CTH = cosine of the angle between the wave normal and the geomagnetic field
DA = calculated differential absorption rate (dB/s)
DAE = an observed value of differential absorption rate (dB/s)
DTR = the number of radians per angular degree
ED = electron concentration (m^{-3})
ELD = elevation angle of the rocket from the transmitter (deg)
F = radio wave frequency of the propagation experiment (Hz)
FLAT = flattening factor of an oblate earth
FR = calculated Faraday rotation rate (deg/s)
FRE = an observed value of Faraday rotation rate (deg/s)
HL = altitude of the rocket above the earth (meters)
R = the distance from the center of the earth to the rocket (meters)
RIO = refractive index, ordinary mode
RIX = refractive index, extraordinary mode
RKTNO = rocket number
RLATD = geodetic latitude of the rocket (deg, north (pos.), south (neg.))
RLNGD = longitude of the rocket (deg, east (pos.), west (neg.))
ROI = wave polarization, ordinary mode, imaginary part
ROR = wave polarization, ordinary mode, real part
RXI = wave polarization, extraordinary mode, imaginary part
RXR = wave polarization, extraordinary mode, real part
S = the angular cyclotron (gyro) frequency (rad/s)
SPH = sine of the geocentric longitude angle of the rocket
ST = sine of the geocentric polar angle of the rocket
TM = time of the rocket shot (year and decimal part of a year)
V = total velocity of the rocket (m/s)
W = angular wave frequency of the propagation experiment (rad/s)
X = the ratio of plasma frequency to wave frequency
Y = ratio of gyro and wave frequencies, a negative number for the electron
Z = the ratio of collision frequency to angular wave frequency

TABLE 2

```
C     ANALYSIS OF ROCKET FR AND DA DATA
C     ENTER PARAMETERS OF OBLATE EARTH
        A=6.378165E6
        A2 =A*A
        A4 = A2*A2
        FLAT=1.-1./298.3
        B2=(A*FLAT)**2
        A2B2=A2*(1.-FLAT**2)
        A4B4=A4*(1.-FLAT**4)
C     ENTER PARAMETERS
        READ (5,50) RKTNO,TM,F
50      FORMAT (F10.3,F10.1,F10.0)
        CALL COEFF(TM)
        W = 6.2831853*F
        DTR = 1.745329E-2
C     INITIALIZE ELECTRON DENSITY AND COLLISION FREQUENCY
        ED=1.E8
        CF=1.E5
C     ENTER VARIABLES
1       READ (5,51) AZD,ELD,HT,V,RLATD,RLNGD,FRE,DAE
51      FORMAT (8F10.4)
C     CONVERT DEGREES TO RADIANS
        AZ = AZD*DTR
        EL = ELD*DTR
        RLNGR = RLNGD*DTR
        RLATR = RLATD*DTR
        SINLA = SIN(RLATR)
20      SINLA2=SINLA*SINLA
        COSLA2=1.-SINLA2
        SPH = SIN(RLNGR)
        CPH = COS(RLNGR)
C     FIND GEOCENTRIC COORDINATES OF ROCKET
        DEN2=A2-A2B2*SINLA2
        DEN=SQRT(DEN2)
        FAC=(((HT*DEN)+A2)/((HT*DEN)+B2))**2
        CT=SINLA/SQRT(FAC*COSLA2+SINLA2)
        R=SQRT(HT*(HT+2.*DEN)+(A4-A4B4*SINLA2)/DEN2)
21      ST=SQRT(1.-CT**2)
C     CALCULATE GEOMAGNETIC FIELD AT ROCKET
        CALL FIELD (R,ST,CT,SPH,CPH,BR,BT,BP,B)
C     TRANSFORM FIELD COMPONENTS, GEOCENTRIC TO GEODETIC.
        SIND=SINLA*ST-SQRT(COSLA2)*CT
        COSD=SQRT(1.-SIND**2)
        BN=-BT*COSD-BR*SIND
        BD=BT*SIND-BR*COSD
        BW=-BP
        S = -1.758803E11*B
        Y=S/W
C     CALCULATE COSINE OF PROPAGATION ANGLE
        CEL = COS(EL)
        CTH = (CEL*COS(AZ)*BN-SIN(EL)*BD-CEL*SIN(AZ)*BW)/B
        TH=ARCOS(CTH)/DTR
C     CALCULATE FR AND DA COEFFICIENTS
        FV = F*V
        FC = 6.004153E-7*FV
        AC = 1.820428E-7*FV
C     ITERATE TO MATCH FR AND DA
        WRITE (6,67)
67      FORMAT (1H1)
```

TABLE 3

```
      WRITE (6,68) RKTNO,TM
68    FORMAT (3X,F10.3,F10.1/)
      I = 6
      DO 2 N=1,I
C   CALCULATE REFRACTION AND ABSORPTION INDICES
      CALL SENWYL (ED,CF,S,CTH,W,RIO,RIX,AIO,AIX,ROR,RXR,ROI,RXI)
      FR = FC*(RIO-RIX)
      DA=AC*ABS(AIX-AIO)
      X = 3182.6018*ED/W/W
      Z=CF/W
      WRITE(6,61)AZD,BN,ROR,RIO,F,HT
      WRITE(6,62)ELD,BD,RXR,RIX,FRE,ED
      WRITE(6,63)RLATD,BW,ROI,AIO,FR,X
      WRITE(6,64)RLNGD,B,RXI,AIX,DAE,CF
      WRITE(6,65)V,Y,CTH,TH,DA,Z
61    FORMAT('     AZ ',1PE13.6,'    BN ',1PE13.6,'   ROR ',1PE13.6,
     &      '    RIO ',1PE13.6,'     F ',1PE13.6,'    HT ',1PE13.6)
62    FORMAT('     EL ',1PE13.6,'    BD ',1PE13.6,'   RXR ',1PE13.6,
     &      '    RIX ',1PE13.6,'   FRE ',1PE13.6,'    ED ',1PE13.6)
63    FORMAT('    LAT ',1PE13.6,'    BW ',1PE13.6,'   ROI ',1PE13.6,
     &      '    AIO ',1PE13.6,'    FR ',1PE13.6,'     X ',1PE13.6)
64    FORMAT('    LNG ',1PE13.6,'     B ',1PE13.6,'   RXI ',1PE13.6,
     &      '    AIX ',1PE13.6,'   DAE ',1PE13.6,'    CF ',1PE13.6)
65    FORMAT('      V ',1PE13.6,'     Y ',1PE13.6,'   COS ',1PE13.6,
     &      '     TH ',1PE13.6,'    DA ',1PE13.6,'     Z ',1PE13.6)
      WRITE (6,60)
60    FORMAT (1H )
      ED=FRE/FR*ED
      CF=FR/FRE*DAE/DA*CF
2     CONTINUE
      GO TO 1
      END
```

Statement (2-11) enters the rocket number, the time of the shot, and the frequency of the radio experiment from a data card.

The subroutine COEFF is called by statement (2-13) to obtain harmonic coefficients representing the geomagnetic field at the time TM. The geomagnetic field is represented by the numerical model of Cain et al. (1967), or Cain (1968). The harmonic coefficients of this model are for epoch 1960.0 and are designated by Cain as "GSFC (12/66)". The coefficients include time derivatives from which coefficients for some other time, TM, are generated by the subroutine COEFF. Subroutine COEFF, as listed in Table 6, is a modified form of a program originally provided by J. C. Cain (Cain, 1968).

Statements (2-17) and (2-18) specify initial values of electron density and collision frequency to begin the iterative process of calculating values of FR and DA which will match the experimentally measured values FRE and DAE, respectively.

Statement (2-20) enters coordinates of the rocket and experimentally measured values of Faraday rate FRE, and differential absorption rate DAE.

Statements (2-32) through (2-38) convert the geodetic coordinates of the rocket to geocentric coordinates for use by the subroutine FIELD which calculates the geomagnetic field components. Subroutine FIELD, as listed in Table 8, is also a modified form of a program originally provided by J. C. Cain (Cain, 1968). The field components are expressed in terms of north, west, and down components; and functions of the total field are computed by statements (2-41) to (2-48).

The numerical coefficient of (1-47) is the charge to mass ratio of the electron (e/m).

The angle between the wave normal and $\vec{B}$ of the earth, required by the Sen-Wyller equations is found by statements (2-49) to (2-52).

To avoid repetitious evaluation, during the iterative process, of the coefficient of the FR and DA equations (Mechtly et al., 1967), the coefficient are evaluated, (2-53) to (2-56), before the DO loop which performs the iterations. The numerical coefficient of (2-55) is 180/c. The numerical coefficient of (2-56) is 2 x (4.3429. . .) x $2\pi/c$.

The DO loop is set for six iterations by (3-3).

Refraction and absorption indices are calculated, (3-6), by the subroutine SENWYL, and finally FR, and DA; and the conventional magnetionic variables X and Z are obtained by statements (3-7) to (3-10). The numerical coefficient of (3-9) is $e^2/\varepsilon_0 m$.

Statements (3-11) to (3-27) print all the parameters and variables of interest for a given iteration.

Statements (3-28) and (3-29) provide more correct values of electron density and collision frequency for the next iteration. These equations follow from first order approximations (Mechtly et al., 1967).

After the specified number of iterations, the program returns by (3-31) to the read statement (2-20) for the next set of data, which usually corresponds to one second later in flight time or about 1.5 km higher in altitude.

3. ANALYSIS OF DIFFERENTIAL-ABSORPTION DATA IN THE ABSENCE OF FARADAY DATA

The program for the analysis of DA data in the absence of FR data is listed in Tables 4 and 5. This program is, for the most part, identical to the program for the simultaneous analysis of FR and DA data. The two programs differ in the following particulars.

Statement (4-17) initializes only ED, and not CF. We assume that the collision frequency model of statement (4-22), CF = CFM * P, is valid at altitudes (below about 85 km) where only DA data are available. The proportionality constant CFM is the value of the ratio CF/P which is judged to best represent the values of CF from the FR and DA program (Tables 1, 2, and 3) and the values of atmospheric pressure P (N/m^2) from a selected CIRA or U.S. Standard Atmosphere model.

Values of pressure are entered from data cards, statement (4-20).

Iteration is done only on the variable DA, (5-26), and not on the variable CF as before, since CF is assumed to be known from the model (4-22).

TABLE 4

```
C     ANALYSIS OF DA WITHOUT FR, GIVEN CF MODEL
C     ENTER PARAMETERS OF OBLATE EARTH
        A=6.378165E6
        A2 =A*A
        A4 = A2*A2
        FLAT=1.-1./298.3
        B2=(A*FLAT)**2
        A2B2=A2*(1.-FLAT**2)
        A4B4=A4*(1.-FLAT**4)
C     ENTER PARAMETERS
        READ (5,50) RKTNO,TM,F,CFM
  50    FORMAT (F10.3,F10.1,F10.0,F10.0)
        CALL COEFF (TM)
        W = 6.2831853*F
        DTR = 1.745329E-2
C     INITIALIZE ELECTRON DENSITY
        ED=1.E7
C     ENTER VARIABLES
  1     READ (5,51) AZD,ELD,HT,V,RLATD,RLNGD,FRE,DAE
        READ (5,51)P
  51    FORMAT (8F10.4)
        CF = CFM*P
C     CONVERT DEGREES TO RADIANS
        AZ = AZD*DTR
        EL = ELD*DTR
        RLATR = RLATD*DTR
        RLNGR = RLNGD*DTR
        SINLA = SIN(RLATR)
  20    SINLA2=SINLA*SINLA
        COSLA2=1.-SINLA2
        SPH = SIN(RLNGR)
        CPH = COS(RLNGR)
C     FIND GEOCENTRIC COORDINATES OF ROCKET
        DEN2=A2-A2B2*SINLA2
        DEN=SQRT(DEN2)
        FAC=(((HT*DEN)+A2)/((HT*DEN)+B2))**2
        CT=SINLA/SQRT(FAC*COSLA2+SINLA2)
        R=SQRT(HT*(HT+2.*DEN)+(A4-A4B4*SINLA2)/DEN2)
  21    ST=SQRT(1.-CT**2)
C     CALCULATE GEOMAGNETIC FIELD AT ROCKET
        CALL FIELD (R,ST,CT,SPH,CPH,BR,BT,BP,B)
C     TRANSFORM FIELD COMPONENTS, GEOCENTRIC TO GEODETIC.
        SIND=SINLA*ST-SQRT(COSLA2)*CT
        COSD=SQRT(1.-SIND**2)
        BN=-BT*COSD-BR*SIND
        BD=BT*SIND-BR*COSD
        BW=-BP
        S = -1.758803E11*B
        Y=S/W
C     CALCULATE COSINE OF PROPAGATION ANGLE
        CEL = COS(EL)
        CTH = (CEL*COS(AZ)*BN-SIN(EL)*BD-CEL*SIN(AZ)*BW)/B
        TH = ARCOS(CTH)/DTR
C     CALCULATE FR AND DA COEFFICIENTS
        FV = F*V
        FC = 6.004153E-7*FV
        AC = 1.820428E-7*FV
C     ITERATE TO MATCH DA
        WRITE (6,67)
  67    FORMAT (1H1)
```

TABLE 5

```
      WRITE (6,68) RKTNO,TM,CFM
68    FORMAT (3X,F10.3,F10.1,1PE10.2/)
      I = 6
      DO 2 N=1,I
      CALL SENWYL (ED,CF,S,CTH,W,RIO,RIX,AIO,AIX,ROR,RXR,ROI,RXI)
      FR = FC*(RIO-RIX)
      DA=AC*ABS(AIX-AIO)
      X = 3182.6018*ED/W/W
      WRITE(6,61)AZD,BN,ROR,RIO,F,HT
      WRITE(6,62)ELD,BD,RXR,RIX,P,ED
      WRITE(6,63)RLATD,BW,ROI,AIO,FR,X
      WRITE(6,64)RLNGD,B,RXI,AIX,DAE,CF
      WRITE(6,65)V,Y,CTH,TH,DA,Z
61    FORMAT('     AZ ',1PE13.6,'     BN ',1PE13.6,'    ROR ',1PE13.6,
     &       '    RIO ',1PE13.6,'      F ',1PE13.6,'     HT ',1PE13.6)
62    FORMAT('     EL ',1PE13.6,'     BD ',1PE13.6,'    RXR ',1PE13.6,
     &       '    RIX ',1PE13.6,'      P ',1PE13.6,'     ED ',1PE13.6)
63    FORMAT('    LAT ',1PE13.6,'     BW ',1PE13.6,'    ROI ',1PE13.6,
     &       '    AIO ',1PE13.6,'     FR ',1PE13.6,'      X ',1PE13.6)
64    FORMAT('    LNG ',1PE13.6,'      B ',1PE13.6,'    RXI ',1PE13.6,
     &       '    AIX ',1PE13.6,'    DAE ',1PE13.6,'     CF ',1PE13.6)
65    FORMAT('      V ',1PE13.6,'      Y ',1PE13.6,'    COS ',1PE13.6,
     &       '     TH ',1PE13.6,'     DA ',1PE13.6,'      Z ',1PE13.6)
      WRITE (6,60)
60    FORMAT (1H )
      ED = ED*DAE/DA
2     CONTINUE
      GO TO 1
      END
```

4. SUBROUTINES FOR CALCULATING THE GEOMAGNETIC FIELD

The subroutine COEFF, listed in Table 6, and FIELD, listed in Table 7, follow directly from FORTRAN programs published by J. C. Cain and his coworkers of NASA's Goddard Space Flight Center. The Data Users' Note NSSDC 68-11 (Cain, 1968) and the paper by Cain *et al*. (1967) give a complete and detailed account of the foundations, development, and listing of programs for computing the geomagnetic field. We do not attempt to recapitulate the vast amount of work underlying these programs in the present report.

TABLE 6

```
      SUBROUTINE COEFF(TM)
      DIMENSION G(11,11),GT(11,11),SHMIT(11,11)
      COMMON /COEFFS/TG(11,11)
C     READ SPHERICAL HARMONIC COEFFICIENTS
C       THE COEFFICIENTS IN THE DATA STATEMENT ARE GSFC(12/66)
C       THIS G ARRAY CONTAINS BOTH G AND H VALUES
      DATA G/
     Z0.,-30401.2,-1540.1,1307.1,949.3,-233.5,49.2,72.2,8.5,10.4,-2.9,
     15778.2,-2163.8,2997.9,-1988.9,803.5,355.7,57.5,-53.7,6.5,5.8,-.9,
     2-1932.,202.9,1590.3,1276.8,502.9,228.4,-.8,7.9,-9.3,7.5,-2.2,
     3-425.4,227.8,-133.8,881.2,-397.7,-28.8,-238.3,15.6,-9.6,-15.1,.8,
     4160.3,-274.3,2.3,-246.6,266.5,-157.9,-1.5,-24.3,-6.1,12.1,-2.8,
     55.1,117.8,-114.8,-108.9,82.4,-62.2,-2.,-3.6,5.5,4.7,6.4,
     6-12.1,104.4,56.6,-23.4,-14.8,-13.3,-108.9,15.5,-8.1,.2,4.7,
     7-53.7,-27.4,-8.1,7.,24.3,-22.5,-21.4,3.6,13.,1.6,-.2,
     85.4,-11.7,4.2,-15.3,4.6,21.9,-.7,-17.1,7.4,.9,1.8,
     9-22.4,13.8,6.3,-3.,-1.9,9.,11.5,.1,-1.5,.2,2.,
     D-.1,4.5,-1.,2.6,-4.4,-1.3,-3.6,4.,1.,-2.,1.1/
      DATA GT/
     Z0.,14.03,-23.29,-.93,1.45,1.61,-.42,-.57,.35,-.10,-.01,
     1-3.71,8.76,-.09,-10.62,.9,.6,.82,-.34,.5,-.13,-.13,
     2-14.31,-16.62,-4.56,2.31,-1.75,3.34,.82,-1.44,1.7,-1.2,.88,
     35.2,2.53,-6.98,-5.89,.66,-.04,2.35,-.9,-.11,.08,-.18,
     4-2.19,-.14,1.88,-6.52,-3.01,-.6,.83,.03,.34,-.08,.17,
     52.24,1.59,-2.61,.5,-.12,1.76,.01,-.6,-.07,-.39,-.02,
     6.05,.09,2.55,-1.19,.33,.84,.23,-.17,.43,-.36,.05,
     7-.96,.01,.43,.75,-.33,.49,.9,-.64,-.15,.47,.17,
     8-.5,-.21,.03,-.79,.05,.1,-.36,-.43,-.42,.37,.16,
     9.66,.54,.03,.35,-.03,-.01,.45,-.05,.75,-.46,.31,
     D-.61,-.64,.02,.05,-.63,-.07,.07,-.03,-.02,-.45,-.23/
C     CALCULATE NORMALIZATION CONVERSION FACTORS
      SHMIT(1,1)=-1.
      MAXN =11
      DO 15 N=2,MAXN
      SHMIT(N,1)=SHMIT(N-1,1)*FLOAT(2*N-3)/FLOAT(N-1)
      SHMIT(1,N)=0.
      JJ=2
      DO 15 M=2,N
      SHMIT(N,M)=SHMIT(N,M-1)*SQRT(FLOAT((N-M+1)*JJ)/FLOAT(N+M-2))
      SHMIT(M-1,N)=SHMIT(N,M)
   15 JJ=1
C     CONVERT COEFFICIENTS, SCHMIDT TO GAUSS
      DO 16 N=2,MAXN
      DO 16 M=1,N
      G(N,M)=G(N,M)*SHMIT(N,M)
      GT(N,M)=GT(N,M)*SHMIT(N,M)
      IF (M.EQ.1) GO TO 16
      G(M-1,N)=G(M-1,N)*SHMIT(M-1,N)
      GT(M-1,N)=GT(M-1,N)*SHMIT(M-1,N)
   16 CONTINUE
C     CONVERT COEFFICIENTS TO NEW TIME
   17 T=TM-1960.0
      DO 18 N=1,MAXN
      DO 18 M=1,N
      TG(N,M)=G(N,M)+T*GT(N,M)
      IF (M.EQ.1) GO TO 18
      TG(M-1,N)=G(M-1,N)+T*GT(M-1,N)
   18 CONTINUE
      RETURN
      END
```

TABLE 7

```
      SUBROUTINE FIELD (R,ST,CT,SPH,CPH,BR,BT,BP,B)
      DIMENSION P(11,11), DP(11,11), CONST(11,11)
      DIMENSION SP(11), CP(11), FN(11), FM(11)
      COMMON/COEFFS/G(11,11)
      NMAX = 11
1     P(1,1) = 1.
      DP(1,1) = 0.
      SP(1) = 0.
      CP(1) = 1.
      DO 2 N = 2,11
      FN(N) = N
      DO 2 M = 1,N
      FM(M) = M-1
2     CONST(N,M) = FLOAT((N-2)**2-(M-1)**2)/FLOAT((2*N-3)*(2*N-5))
3     SP(2) = SPH
      CP(2) = CPH
      DO 4 M = 3,NMAX
      SP(M) = SP(2)*CP(M-1)+CP(2)*SP(M-1)
4     CP(M) = CP(2)*CP(M-1)-SP(2)*SP(M-1)
      AOR=6.3712E6/R
      AR = AOR**2
      BT = 0.
      BP = 0.
      BR = 0.
      DO 8 N = 2,NMAX
      AR = AOR*AR
      DO 8 M = 1,N
      IF(M.EQ.N) GO TO 5
      P(N,M)=CT*P(N-1,M)
      DP(N,M)=CT*DP(N-1,M)-ST*P(N-1,M)
      IF(M.EQ.N-1) GO TO 7
      P(N,M)=P(N,M)-CONST(N,M)*P(N-2,M)
      DP(N,M)=DP(N,M)-CONST(N,M)*DP(N-2,M)
      GO TO 7
5     P(N,N) = ST*P(N-1,N-1)
      DP(N,N) = ST*DP(N-1,N-1)+CT*P(N-1,N-1)
7     PAR = P(N,M)*AR
      IF (M.EQ.1) GO TO 9
      TEMP = G(N,M)*CP(M)+G(M-1,N)*SP(M)
      BP = BP-(G(N,M)*SP(M)-G(M-1,N)*CP(M))*FM(M)*PAR
      GO TO 10
9     TEMP = G(N,M)*CP(M)
      BP = BP-(G(N,M)*SP(M))*FM(M)*PAR
10    BT = BT+TEMP*DP(N,M)*AR
8     BR = BR-TEMP*FN(N)*PAR
      BP = BP/ST
      B = SQRT(BT*BT+BP*BP+BR*BR)
      GTT=1.E-9
      BR=BR*GTT
      BT=BT*GTT
      BP=BP*GTT
      B=B*GTT
      RETURN
      END
```

5. A SUBROUTINE FOR GENERALIZED MAGNETOIONIC CALCULATIONS

In the region of the lower ionosphere between about 50 and 90 km, the collisions of electrons with neutral molecules appreciably influence the refraction and absorption indices of radio waves which are suitable for investigating this region. The proportionality of electron collision frequency and electron kinetic energy (Phelps and Pack, 1959) must be an intrinsic part of the equations for the numerical analysis of experimental data from this region. The Sen-Wyller equations (Sen and Wyller, 1960) are a convenient formulation of the generalized magnetoionic theory which includes the energy dependence of collision frequency.

Subroutine SENWYL, listed in Table 8, is a statement of the Sen-Wyller equations.

Statements (8-5) to (8-10) are polynomial approximations (Hara, 1963) of the semiconductor integrals appropriate for the ionosphere. Hara asserts that these approximations are in error by less than 1% at most.

Statement (8-11) calculates the square of the angular plasma frequency, P2. The numerical coefficient is $e^2/\varepsilon_0 m$.

Convenient intermediate variables are calculated by (8-12) to (8-19).

Statements (8-20) to (8-25) are equations (55) from the paper by Sen and Wyller (1960).

Statements (8-26) to (8-28) are equation (56) of Sen and Wyller; (8-29) to (8-33) are equation (28); (8-34) to (8-44) are equation (27); and (8-45) to (8-49) separate the real and imaginary parts of the complex refractive indices for the two magnetoionic modes.

Wave polarizations are calculated by (8-50) to (8-57). These statements correspond to equation (35) of Sen and Wyller.

TABLE 8

```
      SUBROUTINE SENWYL(ED,CF,S,CPH,W,RIO,RIX,AIO,AIX,ROR,RXR,ROI,RXI)
C     GENERALIZED MAGNETO-IONIC THEORY, SEN-WYLLER EQUATIONS.
      COMPLEX EI,EII,EIII,AA,BB,CC,DD,EE,VC,VD,VE,VF,VG,VH,VO,VX
      COMPLEX CMPLX,CSQRT,BBVA,CCCPH,RO,RX
C     DEFINE BURKE-HARA FUNCTIONS FOR C3/2 AND C5/2.
      C32(X) = (X*(X*(X*(X+24.653115)+113.9416)+11.287513)+.023983474)/(
     1X*(X*(X*(X*(X*(X+24.656819)+120.49512)+289.58085)+149.21254)+9.387
     27372)+.018064128)
      C52(X) = (X*(X*(X+6.6945939)+16.901002)+1.1630641)/(X*(X*(X*(X*(X+
     16.6314497)+35.355257)+68.920505)+64.093464)+4.3605732)
      P2 = 3182.6018*ED
      WPS = W+S
      WMS = W-S
      Q = P2/W/CF
      R = Q/CF
      T = 2.5*Q
      A1 = W/CF
      A2 = WMS/CF
      A3 = WPS/CF
      A = R*W*C32(A1)
      B = T*C52(A1)
      C = R*WMS*C32(A2)
      D = T*C52(A2)
      E = R*WPS*C32(A3)
      F = T*C52(A3)
      EI = CMPLX (1.-A,-B)
      EII = CMPLX (.5*(F-D),.5*(C-E))
      EIII = CMPLX (A-.5*(C+E),B-.5*(F+D))
      AA = 2.*EI*(EI+EIII)
      BB  = EIII*(EI+EIII)+EII**2
      CC = 2.*EI*EII
      DD = 2.*EI
      EE = 2.*EIII
      VB = CPH*CPH
      VA = 1.-VB
      VC = BB*BB*VA*VA-CC*CC*VB
      VD=CSQRT(VC)
      BBVA=BB*VA
      VE=AA+BBVA
      VF = DD+EE*VA
      VG = (VE+VD)/VF
      VH = (VE-VD)/VF
      VO=CSQRT(VG)
      VX=CSQRT(VH)
C     SEPARATE INDICES
      RIO = REAL (VO)
      AIO = -AIMAG(VO)
      RIX = REAL (VX)
      AIX = -AIMAG(VX)
C     CALCULATE WAVE POLARIZATIONS
      CCCPH=CC*CPH
      RO=-((BBVA-VD)/CCCPH)
      ROR=REAL(RO)
      ROI=AIMAG(RO)
      RX=-((BBVA+VD)/CCCPH)
      RXR=REAL(RX)
      RXI=AIMAG(RX)
      RETURN
      END
```

6. RAY TRACING FOR THE ANALYSIS OF STANDING WAVES

6.1 Haselgrove Method

The Haselgrove (1954) method of ray tracing which provides wave-normal directions and ray coordinates at the end of each integration step is used, with some modification.

The coordinates of the ray are denoted by y(1), y(2), y(3) and the direction cosines of the wave normal by AP(1), AP(2), AP(3) in the Cartesian coordinate system. The coordinate system is oriented such that y(3) is the vertical axis and the wave normal is in the y(1), y(3) plane, this will be called the propagation plane. See Figure 6.1.

The propagation plane [y(1), y(3)] makes an angle AZI with magnetic north and DIP is the magnetic dip. AZI is determined by the plane of the rocket trajectory. The direction cosines of B are then:

$$B(1) = \cos DIP \; . \; \cos AZI$$

$$B(2) = \cos DIP \; . \; \sin AZI$$

$$B(3) = - \sin AZI$$

Both DIP and AZI are input parameters for the ray-tracing program.

6.2 Differential Equations of the Ray

The ray paths are calculated by integrating the following simultaneous differential equations

$$\frac{du_i}{dt} = \frac{\partial G}{\partial y_i} \qquad (1)$$

$$\frac{dy_i}{dt} = \frac{\partial G}{\partial U_i} \qquad (2)$$

(In the FORTRAN program u_i is denoted by Y(I + 3))



Figure 6.1 Orientation of the coordinate system for the ray-tracing problem.

The parameter, G, in these equations is used in defining the normal index surface. This surface is constructed about a point, Q (say), in the medium by first drawing a vector from Q in the direction of a wave normal. Then the locus of the point on the vector which is displaced from Q by a distance equal to the phase refractive index, μ, generates the normal index surface. In deriving the equations Haselgrove (1954) defined the normal index surface by

$$G(y_i, u_i) = 1$$

where
$$G \equiv \frac{u}{\mu} \equiv \frac{(u_1^2 + u_2^2 + u_3^2)^{1/2}}{\mu(y_i, u_i)} \equiv 1 \qquad (3)$$

Thus the end of the vector, $\vec{u}$ (which is in the direction of wave normal), generates the normal index surface since the condition, $\frac{u}{\mu} = 1$, implies that the magnitude of $\vec{u}$ is equal to μ.

In equations (1) and (2), t is a variable of integration and, in the sense used here, it has the dimension of length. The significance of this variable is such that if, in a small interval of time, the phase of the radio wave advances a distance dt in free space, then in the medium it advances a distance ds = $(dy_1^2 + dy_2^2 + dy_3^2)^{1/2}$ in the direction of the ray, where dy_i are given by equation (2).

Neglecting collisions and using equation (3) and

$$D = 2(1-X) - y_T^2 \pm \{y_T^4 + 4y_L^2(1-X)^2\}^{1/2}$$

$$\mu^2 = 1 - \frac{2X(1-X)}{D}$$

$$\cos\theta = \frac{1}{u}\sum_{i=1}^{3} u_i\, b_i,$$

and
$$\sin\theta = (1 - \cos^2\theta)^{1/2} \text{ (always positive)},$$

$$\frac{dy_i}{dt} = \frac{1}{\mu}\left[\frac{u_i}{\mu} - F1(b_i - \cos\theta\,\frac{u_i}{u}\right] \qquad (4)$$

$$\frac{du_i}{dt} = - F_2 \frac{\partial X}{\partial y_i} \qquad (5)$$

where
$$F_1 = \frac{2X(1-X)}{\mu^2 D^2} \{1 + \frac{2(1-X)^2 - y_T^2}{D - 2(1-X) + y_T^2}\} y y_L$$

and
$$F_2 = \frac{u}{D\mu^3} \left[1 - 2X + \frac{2X(1-X)}{D} \{1 + \frac{2y_L^2(1-X)}{D - 2(1-X) + y_T^2}\} \right]$$

Thus (4) and (5) are a set of six differential equations. They are integrated simultaneously to obtain the parameters of the ray path.

6.3 Method of Integrating

The method of integrating equations (4) and (5) is the Runge-Kutta fourth order process as modified by Gill (1951).

The set of six simultaneous differential equations are written in generalized form:

$$\frac{dy_i}{dt} = f_i(y_1, y_2, y_3, y_4, y_5, y_6) \quad (i = 1, 2, \ldots 6).$$

If the initial values of y_i are y_{io} then four approximations to the new values corresponding to an interval h in t are

$$k_{i0} = hf_i(y_{10}, y_{20}, \ldots y_{60})$$
$$k_{i1} = hf_i(y_{11}, y_{21}, \ldots y_{61})$$
$$k_{i2} = hf_i(y_{12}, y_{22}, \ldots y_{62})$$
$$k_{i3} = hf_i(y_{13}, y_{23}, \ldots y_{63})$$

in which
$$y_{i1} = y_{i0} + k_{i0}/2$$
$$y_{i2} = y_{i1} + (1 - \sqrt{1/2})\,(k_{i1} - q_{i1})$$

where
$$q_{i1} = k_{i0}$$
$$y_{i3} = y_{i2} + (1 + \sqrt{1/2})\,(k_{i2} - q_{i2})$$

where
$$q_{i2} = (2 - \sqrt{2})\,k_{i1} - (2 - 3\sqrt{1/2})\,q_{i1}.$$

The adopted value of y_i at the end of the step is then

$$y_{i4} = y_{i3} + 1/6\ k_{i3} - 1/3\ q_{i3}$$

where
$$q_{i3} = (2 + \sqrt{2})\ k_{i2} - (2 + 3\sqrt{1/2})\ q_{i2}.$$

To economize in the computer storage requirements, a different notation is used in the FORTRAN program. Suppose that we are at the nth stage of approximations, that is y_{ij}, k_{ij}, q_{ij} have been calculated for $j = n - 1$ and they are to be calculated for $j = n$. In the FORTRAN program (y_{ij}, k_{ij}, q_{ij}, $i = 1, 2 \ldots, 6$, $j = n-1$) are denoted by (AY(I), AK(I), AQ(I), I = 1, 2, . ., 6) and (y_{ij}, k_{ij}, q_{ij}, $i = 1, 2, \ldots 6$, $j = n$) are denoted by (Y(I), K(I), Q(I), I = 1, 2, . ., 6).

6.4 Initial Conditions for the Ray

Let A be the take-off angle of the ray then the initial direction cosines of the wave normal are:

$$\mathrm{AP}(1) = \cos A$$

$$\mathrm{AP}(2) = 0$$

$$\mathrm{AP}(3) = \sin A$$

The coordinates (y_1, y_2, y_3) at the base height of the ionosphere are called (H_1, H_2, H_3) and are given by $(\frac{HB}{\tan A}, 0, HB)$.

Because of the method of integration used, ray tracing starts at a distance Q (taken as 1 km; an input parameter to the program) beyond HB in the direction of the wave normal. Thus the initial coordinates of the ray are:

$$y(i) = H(i) + Q \cdot \mathrm{AP}(i) \qquad i = 1, 2, 3$$

6.5 Ray Tracing near the Reflection Height

The refractive index μ and hence the parameters $y(i + 3) = \mu\ \mathrm{AP}(i)$ vanish rapidly near the reflection height, therefore the accumulated errors in the calculated values of y(4), y(5), y(6) would become serious. Therefore, the tracing of the upgoing ray is turncated when it reaches within 0.2 km of the reflection height. Then the program starts tracing the downgoing ray. The initial position y(1), y(2), y(3) of the downgoing ray is taken to be the position of the upgoing ray at truncation. The initial values for y(4), y(5), y(6) of the downgoing ray are determined from the roots of the Booker quartic equation, Budden (1961).

6.6 The Computer Program

The computer program is written in FORTRAN IV and consists of a main program and four subroutines. The main program includes the integration instructions. The first subroutine returns the initial coordinates of the ray and the direction-cosines of the wave normal at the base of the ionosphere. The second subroutine returns the values of X and $\frac{\partial X}{\partial y(i)}$ at each point of the integration, the third solves the Booker quartic equation and returns its roots, and the fourth increments the take-off angle at the end of each integration and senses when all rays have been computed.

A simplified form of the flow diagram of the program is shown in Figure 6.2. Constant parameters are defined first (9-3) to (9-18), next, frequency and the magnetic field data are read (9-19), and then the program branches to the subroutine DETXP. On entering this subroutine the first time, the parameters describing the ionosphere are read and these instructions, (13-6) to (13-41), are by-passed on succeeding occasions. DETXP also determines the initial direction cosines of the wave normal and the starting coordinates of the ray at the level where y(3)=HB, (13-47) to (13-53). These values are returned to the master program which then determines the starting point for the integration. The values Y and Y_L and Y_T are then determined (9-47) to (9-53), before branching to the subroutine DETX which returns the values of X and its spatial derivatives. This information then allows the calculation of the refractive index, AMU from the Appleton-Hartree equations, (9-56) to (10-20). The program then proceeds through a loop involving the



Figure 6.2 Simplified flow diagram for the ray tracing computer program.

TABLE 9

```
C OBLIQUE INCIDENCE RAY-TRACING
      DIMENSION B(3),AP(3),Y(8),AY(8),AK(8),AQ(8),H(3),DX(3),W(8)
C ENTER CONSTANTS FOR R-K-G SOLUTIONS
      ST=SQRT(2.)
      S1=1.-1./ST
      S2=1.+1./ST
      S3=2.-ST
      S4=2.-3./ST
      S5=2.+ST
      S6=2.+3./ST
      PI2=6.2831854
C ENTER PARAMETERS
      CO=PI2/360.
      Q= 1.0
      SG = 0.5
      JP = 8.0
      M = 1.0
      JF = 4.0
1     READ(5,2) F,FH,DIP,AZI
2     FORMAT(F7.0,F8.0,F5.0,F4.0)
      WRITE(6,3) F,FH,DIP,AZI
 3    FORMAT(1H1,6HFREQ =,-6PF6.3,10X8HG.FREQ =,-6PF6.3,10X5HDIP =,0PF5.
     11,10X10HAZIMUTH = ,F4.0)
      READ (5,85) SE,SF
85    FORMAT(2F6.0)
C FIND FIELD DIRECTION COSINES
      DIP=DIP*CO
      AZI=AZI*CO
      BY=-FH/F
      C=COS(DIP)
      B(1)=C*COS(AZI)
      B(2)=C*SIN(AZI)
      B(3)=-SIN(DIP)
      J=0
C START POINT FOR EACH RAY ELEMENT
    4 CALL DETXP(AP,H,J,IOX,HB,HR,F,PI2,CO,BY)
      IM=JP
C FIND INITIAL VALUES OF Y1 TO Y3
      DO 20 I=1,3,1
   20 Y(I)=H(I)+Q*AP(I)
      S=AP(1)
      CF=AP(3)
      K=0
      L=1
      SD=SG
C FIND Y,YL,YT
    5 C=0.
      DO 6 I=1,3,1
    6 C=C+AP(I)*B(I)
      YL=BY*C
      SC=SQRT(1.-C*C)
      YT=BY*SC
      YT2=YT*YT
C FIND X AND DX/DH
      CALL DETX(X,DX,Y,SD,K,L)
C FIND INDICES FROM A-H EQ
      X1=1.-X
      X2=2.*X1
      X3=YL*YL*X2*X2
```

TABLE 10

```
      X4=X2-YT2
      X5=2.*X*X1
      X6=SQRT(YT2*YT2+X3)
      DEX=X4-X6
      IF(IOX)7,7,8
    7 D=X4+X6
      IF(K-1) 70,70,9
 70   IF(SF-Y(3))89,89,90
 89   IF(IM-JF)9,10,10
 90   IF(IM-JP)9,10,10
 10   AMUE=1.-X5/DEX
      IF(AMUE)11,11,12
 11   AMUE=-1.
      GO TO 9
 12   AMUE=SQRT(AMUE)
      GO TO 9
    8 D=DEX
    9 AMU=1.-X5/D
      IF(AMU)13,13,14
   14 AMU=SQRT(AMU)
      IF(K-1)15,16,17
   15 IF(L)18,18,19
C PRINT OUTPUT COLUMN HEADINGS
 19   WRITE(6,21)
   21 FORMAT(5X1HX,10X1HY,11X1HH,9X2HMU,10X1HU,9X2HEL,7X5HTHETA,6X5HRATI
     1O,5X6HINCLIN,7X3HMUX)
      L=0
 96   U=AMU
C FIND INITIAL VALUES OF Y4 TO Y6
      DO 22 I=1,3,1
   22 Y(I+3)=AMU*AP(I)
C PRESERVE ALL INITIAL VALUES OF Y
      DO 23 I=1,6,1
   23 AY(I)=Y(I)
      K=1
      GO TO 24
 18   IM=JF
      GO TO 96
   16 IF(L-1)24,24,25
   25 L=0
      GO TO 17
C CALCULATE STEP SIZE
   24 DN=AMU+2.*((1.-AMU*AMU)*(1.+(1.-X*X)*YL*YL/(D-X4))-X*X)/D/AMU
      DN=SG/DN
      IF(SF-Y(3))87,87,88
 87   IF(IM-JF)13,76,76
 88   IF(IM-JP)13,76,76
C PREPARE RESULTS FOR OUTPUT
 76   EL=ATAN(Y(3)/Y(1))/CO
      THE=ATAN(Y(6)/Y(4))/CO
      P1=YT2/X2
      P2=SQRT(YT2*YT2/X2/X2+YL*YL)
      IF(IOX)55,55,56
   55 P1=(P2-P1)/YL
      GO TO 57
   56 P1=-(P2+P1)/YL
   57 R=2.*P1/(1.+P1*P1)
      R=(1.-SQRT(1.-R*R))/R
      AIN = 1.
      IF (IOX) 26,26,27
```

determination of the appropriate values of AK_{in} (11-52) to (11-57), AQ_{in} and Y_{in} (11-59) to (12-3) and (12-30) to (12-42), until the integration step is complete. Each integration step requires the program to run through this loop four times; during this cycle each new value of ray elevation, y(3), is compared with the ordinary-wave reflection height (12-11) if the former is higher the integration increment (DN) is reduced by 1/2, and the integration step is redone (12-5) to (12-11). When the integration step is complete the various parameters of the ray are printed out and the program then determines if the ray-tracing procedure has gone far enough (11-9) and (11-12), in which case the integration is terminated. If the ray is not complete, the ray tracing is continued in this manner until the ray is traced up to within 0.2 km of the ordinary-wave reflection height, whereupon the upgoing ray is truncated (11-10) to (11-11); then a downgoing ray is traced starting from where the upgoing ray was truncated. The initial values of the refractive index and wave normal direction cosines of the downgoing ray are determined from the roots of the Booker quartic equation. The parameters of the Booker quartic equation are determined first (11-14) to (11-30), then the program branches to the subroutine QUAR (11-31) which returns the four roots of the quartic equation, and finally the refractive index, y(4) and y(6) are calculated from the proper root of the B.Q. equation (11-32) to (11-47). Then the program goes back to the integration cycle.

When a ray is complete the program branches to the subroutine REP. This increments the take-off angle so that another ray can be computed after branching back to DETXP. REP also senses when all rays are computed and the program is stopped at this point.

For the purposes of ray tracing the electron density is assumed to be a function of height only, so that in equation (5) $\frac{\partial X}{\partial y(1)}$ and $\frac{\partial X}{\partial y(2)}$ are zero. The

TABLE 11

```
26     WRITE(6,28) (Y(I),I=1,3),AMU,U,EL,THE,R,AIN,AMUE
C PRINT ORD-MODE RESULTS
28     FORMAT(F8.1,F12.1,F11.1,2F11.3,F10.1,F11.1,F11.2,F11.1,F12.3)
C PRINT X-MODE RESULTS
       GO TO 75
27     WRITE(6,29) (Y(I),I=1,3),AMU,U,EL,THE,R,AIN
29     FORMAT(F8.1,F12.1,F11.1,2F11.3,F10.1,F11.1,F11.2,F11.1)
75     IM=1.
13     IF(SE-Y(1))33,86,86
86     IF(HR-Y(3)-.2)30,30,31
    31 IF(Y(6))32,17,17
    32 IF(HB-Y(3)+.5)17,17,33
30     IF(Y(6))17,17,34
34     Y2=BY*BY
       T1=1.-Y2
       T2=B(3)*B(3)*Y2-1.
       T3A=2.*B(1)*B(3)*Y2
       S9=S*S
       C2=CF*CF
       T3=T3A*S
       T4=Y2*(1.-B(3)*B(3)*C2+S9*B(1)*B(1))
       T5=-T3A*C2*S
       T6=B(1)*B(1)*S9*C2*Y2
       V=C2-X
C FIND COEFFS OF QUARTIC
       AL=T1+X*T2
       BE=T3*X
       GA=-X2*V+2.*Y2*V+X*T4
       DE=T5*X
       EP=X1*V*V-C2*Y2*V-T6*X
       CALL QUAR(AL,BE,GA,DE,EP,W)
       I=2
102    IF(W(I)) 100,101,100
101    I=I+2
       IF(I-8)102,102,103
103    QU=W(3)
       GO TO 107
100    I=2
106    IF(W(I))104,105,104
104    I=I+2
       GO TO 106
105    QU=W(I-1)
107    TT=-S/QU
       TT1=SQRT(TT*TT+1.)
       AMU=SQRT(QU*QU+S*S)
       Y(4)=AMU*TT/TT1
       Y(6)=-AMU/TT1
       L=0
       K=0
       GO TO 35
C PART OF RT MEMBER OF EQ 4
 17    F1=X*X2*(1.+(X1*X2-YT2)/(D-X4))*BY*YL/AMU/AMU/D/D
C PART OF RT MEMBER OF EQ 5
       F2=(1.-2.*X+X2*X*(1.+X3/X2/(D-X4))/D)*U/D/AMU/AMU/AMU
       DO 36 I=1,3,1
       AK(I)=(AP(I)-F1*(B(I)-C*AP(I)))/AMU*DN
    36 AK(I+3)=-F2*DX(I)*DN
       GO TO(37,38,39,40),K
C FIND Y1-Y6 IN FIRST APPROX
37     DO 41 I=1,6,1
```

TABLE 12

```
       Y(I)=Y(I)+AK(I)/2.
    41 AQ(I)=AK(I)
    50 K=K+1
    53 IF(Y(3)-HB)42,43,43
    42 DO 44 I=1,8,1
    44 Y(I)=AY(I)
       DN=DN/2.
       K=1
       L=2
       GO TO 35
    43 IF(Y(3)-HR)35,42,42
    35 U=0.
       DO 45 I=4,6,1
    45 U=U+Y(I)*Y(I)
       U=SQRT(U)
       DO 46 I=1,3,1
    46 AP(I)=Y(I+3)/U
       IF(K-1)5,47,5
  47   BQ=0.
       DO 84 I=1,3,1
  84   BQ=BQ+(Y(I)-AY(I))*(Y(I)-AY(I))
       IF(BQ)95,64,95
  95   SD=(Y(3)-AY(3))*DN/SQRT(BQ)
       IF(SD)63,64,64
 63    SD=-SD
 64    DO 48 I=1,6,1
    48 AY(I)=Y(I)
       IM=IM+1
       GO TO 5
C FIND Y1-Y6 IN SECOND APPROX
 38    DO 49 I=1,6,1
       Y(I)=Y(I)+S1*(AK(I)-AQ(I))
    49 AQ(I)=S3*AK(I)-S4*AQ(I)
       GO TO 50
C FIND Y1-Y6 IN THIRD APPROX
 39    DO 51 I=1,6,1
       Y(I)=Y(I)+S2*(AK(I)-AQ(I))
    51 AQ(I)=S5*AK(I)-S6*AQ(I)
       GO TO 50
C FIND Y1-Y6 IN FOURTH APPROX
 40    DO 52 I=1,6,1
    52 Y(I)=Y(I)+AK(I)/6.-AQ(I)/3.
       K=1
       GO TO 53
 33    CALL REP(J,M)
       GO TO(1,4,54),J
    54 CONTINUE
       END
```

TABLE 13

```
      SUBROUTINE DETXP(AP,H,J,IOX,HB,HR,F,PI2,CO,Y)
      DIMENSION AP(3),H(3),AH(200),X(200),DX(200),AX(200)
      COMMON A,AMAX,DA,AH,X,DX
      IF(J)1,1,2
C DO 1 TO 16 FIRST TIME ONLY
    1 J=1
C FACTOR TO CONVERT N TO X
      CON=1.6021E-19*1.7588E11/8.854E-12/PI2/PI2/F/F*1.E6
      IOX = 0
      READ (5,3) AMIN,AMAX,DA,N
3     FORMAT(F4.0,F5.0,F5.0,I2)
      AMIN=CO*AMIN
      AMAX=CO*AMAX
      DA=CO*DA
      READ(5,4) (AH(I),X(I), I=1,N)
    4 FORMAT(4(F6.0,E9.0))
C CONVERT N TO X
      DO 5 I=1,N,1
5     X(I)=X(I)*CON
      N=N-1
      DO 6 I=1,N,1
C FIND CHANGE IN X BETWEEN DATA POINTS
    6 DX(I)=(X(I+1)-X(I))/(AH(I+1)-AH(I))
      HB=AH(1)
      IF(IOX)7,7,8
    7 B=1.
      WRITE (6,9)
 9    FORMAT(3X5HO-RAY)
      GO TO 10
    8 B=1.+Y
      WRITE(6,11)
 11   FORMAT(3X5HX-RAY)
   10 I=1
   15 IF(X(I)-B)12,13,14
   12 I=I+1
      GO TO 15
   13 HR=AH(I)
      GO TO 16
   14 I=I-1
      HR=(B-X(I))/DX(I)+AH(I)
   16 A=AMIN
C ABOVE 1 TO 16 FIRST TIME ONLY
    2 A1=A/CO
      WRITE(6,17)  A1
 17   FORMAT(3X16HTAKE-OFF ANGLE =,F5.1//)
C DIRECTION COSINES OF RAY NORMAL
      AP(1)=COS(A)
      AP(2)=0.
      AP(3)=SIN(A)
      TA=AP(3)/AP(1)
      H(1)=HB/TA
      H(2)=0.
      H(3)=HB
      RETURN
      END
```

TABLE 14

```
      SUBROUTINE DETX(X,DX,Y,SD,I,L)
C FIND X AND DX/DH
      DIMENSION DX(3),Y(6),AH(200),AX(200),ADX(200)
      COMMON A,AMAX,DA,AH,AX,ADX
      IF(I-1)1,2,3
 1    IF(L)4,4,5
 5    X1=0.
      J=2
      DX(3)=AX(J)/(AH(J)-AH(1))
      H2=Y(3)
      H1=AH(1)
      J1=J
 3    X=X1+DX(3)*(Y(3)-H1)
 4    DX(1)=0.
      DX(2)=0.
      RETURN
 2    J=1
 8    IF(AH(J)-Y(3))6,6,7
 6    J=J+1
      GO TO 8
 7    IF(H2-Y(3))9,9,10
 9    H2=Y(3)
      IF(J-J1)11,12,13
 11   J=J1
      GO TO 3
 12   IF(AH(J)-Y(3)-SD)14,3,3
 14   J=J+1
 13   M=J
 16   J1=J
      X=X1+DX(3)*(Y(3)-H1)
      X1=X
      DX(3)=(AX(M)-X)/(AH(M)-Y(3))
      H1=Y(3)
      GO TO 4
 10   H2=Y(3)
      IF(J-J1)20,17,11
 20   IF(J-1)3,3,15
 15   M=J-1
      GO TO 16
 17   IF(Y(3)-AH(J-1)-SD)18,3,3
 18   IF(J-1)3,3,19
 19   J=J-1
      GO TO 15
      END
```

TABLE 15

```
      SUBROUTINE QUAR(A,B,C,D,E,S)
      DIMENSION S(8),QR(2),RR(2)
      PI=3.1415927
      PI2=PI*2.
      A1=B/A
      B1=C/A
      C1=D/A
      D1=E/A
      J= 1
      A2=A1*A1
      PA=B1-3.*A2/8.
      QA=C1-A1*B1/2.+A1*A2/8.
      RA=D1-A1*C1/4.+B1*A2/16.-3.*A2*A2/256.
      P=1.
      Q=-PA/3.
      R=-4.*RA/3.
      W=4.*PA*RA-QA*QA
      AQ=P*R-Q*Q
      AR=(3.*P*Q*R-P*P*W)/2.-Q*Q*Q
      Z=AR*AR+AQ*AQ*AQ
      IF(AQ)71,72,72
   71 TH=AR/SQRT(-AQ*AQ*AQ)
      IF(TH)75,76,76
   75 TH=-TH
   76 IF(Z)73,74,74
   73 TH=ATAN(SQRT(1.-TH*TH)/TH)
      TH=TH/3.
   81 Y1=2.*SQRT(-AQ)*COS(TH)
      IF(AR)77,78,78
   77 Y1=-Y1
   78 Y1=(Y1-Q)/P
      IF(Y1-PA)79,80,80
   79 TH=TH+PI2/3.
      J=J+1
      IF(J-3)81,81,80
   74 TG=TH+SQRT(TH*TH-1.)
      TH = ALOG(TG)
      TG=TH/3.
      TH=-TG
      Y1=SQRT(-AQ)*(EXP(TG)+EXP(TH))
      IF(AR)82,83,83
   82 Y1=-Y1
   83 Y1=(Y1-Q)/P
      GO TO 80
   72 TH=AR/SQRT(AQ*AQ*AQ)
      IF(TH)86,87,87
   86 TH=-TH
   87 TG=TH+SQRT(TH*TH+1.)
      TH = ALOG(TG)
      TG=TH/3.
      TH=-TG
      Y1=SQRT(AQ)*(EXP(TG)-EXP(TH))
      IF(AR)84,85,85
   84 Y1=-Y1
   85 Y1=(Y1-Q)/P
   80 TH=SQRT(Y1-PA)
      QR(1)=TH
      QR(2)=-TH
      RR(1)=Y1/2.-QA/2./TH
      RR(2)=Y1/2.+QA/2./TH
```

TABLE 16

```
      DO 99 I=1,2,1
      SR=QR(I)*QR(I)-4.*RR(I)
      IF(SR)57,58,58
   58 ST=0.
      CT=1.
      R=SQRT(SR)
      GO TO 59
   57 ST=1.
      CT=0.
      R=SQRT(-SR)
   59 IF(I-1)96,96,97
   96 J=1
      GO TO 98
   97 J=5
   98 S(J)=-QR(I)/2.+R*CT/2.-A1/4.
      S(J+1)=R*ST/2.
      S(J+2)=-QR(I)/2.-R*CT/2.-A1/4.
   99 S(J+3)=-R*ST/2.
   32 J=1
   33 IF(S(J)-S(J+2))30,30,31
   31 P=S(J)
      S(J)=S(J+2)
      S(J+2)=P
      P=S(J+1)
      S(J+1)=S(J+3)
      S(J+3)=P
      GO TO 32
   30 J=J+2
      IF(J-5)33,33,34
   34 RETURN
      END
```

TABLE 17

```
      SUBROUTINE REP(J,M)
      COMMON A,AMAX,DA
      A=A+DA
      IF(A-AMAX)2,2,1
    1 IF(M)3,3,4
    3 J=1
      GO TO 5
    4 J=3
      GO TO 5
    2 J=2
5     WRITE(6,6)
    6 FORMAT(1H1)
      RETURN
      END
```

electron densities are provided to the subroutine DETXP at a series of heights and between these heights the electron density is assumed to vary linearly.

Input data for the program includes, besides the electron-density parameters and operating frequency (F), parameters describing the magnetic field, of which any variation with height is neglected. These field parameters are the electron gyrofrequency (FH), dip angle (DIP) and the azimuth angle (AZI) which determines the propagation plane. A parameter (SG) which determines the size of each integration step is also supplied and this is selected so that, in general, an integration step in free space would be about 0.5 km. However, this parameter is multiplied by μ/n' at each point in the integration; multiplying by μ would give approximately equal integration steps and further dividing by n' gives shorter integration steps in regions where the refractive index is changing more rapidly.

Figure 6.3 represents the ray-tracing results for a typical rocket shot, the rocket path and only portions of the rays which are of interest for the standing-wave analysis are shown. The standing-wave analysis formulated in section 6 requires the propagation angles (of both the upgoing and downgoing rays) for a number of points on the rocket trajectory. An important feature of the ray-tracing results is that the propagation angles vary very slowly as a function of position. Hence, when the rays are shifted in position due to ray-tracing errors, the corresponding error in propagation angles and hence the error in electron-density profile is small.

The standing-wave analysis, formulated from the Booker quartic equation, Section 8, requires a family of take-off angles of the upgoing and downgoing rays (α and β in Figure 6.3). It should be noted frcm the figure that these take-off angles also vary slowly as functions of position.



Figure 6.3 Rays traced for a series of take-off angles. Only those sections of the rays which intersect the rocket trajectory are shown. (Nike-Apache 14.143).

7. STANDING-WAVE ANALYSIS, FORMULATED FROM THE APPLETON-HARTREE EQUATION

7.1 Method of Analysis

Monro et al. (1968) have described a method of determining electron densities from standing-wave patterns between the X-ray and O-ray reflection levels. Two waves of the same frequency but differing in wavelength are considered in the r_1 and r_2 directions, Figure 7.1. When the X - Y axes are oriented such that $\phi_1 = \tan^{-1} \frac{\mu_1 - \mu_2 \cos\theta}{\mu_2 \sin\theta}$ (or equivalently $\cos \phi_1 = \frac{\mu_1 \sin\theta}{\mu'}$), then the planes of maximum (or minimum) amplitude are normal to Y, the separation between adjacent planes being

$$d = \frac{\lambda_o}{\mu'} \tag{1}$$

where λ_o = free space wavelength

μ_1 = refractive index for the direct ray

μ_2 = refractive index for the reflected ray

$\mu' = (\mu_1^2 + \mu_2^2 - 2\mu_1 \mu_2 \cos \theta)^{1/2}$.

The distance between standing wave minima observed along the rocket trajectory is given by

$$d' = \frac{d}{\cos \phi_3} \tag{2}$$

From the geometry of Figure 7.1, ϕ_3 is determined by:

$$\phi_1 = \tan^{-1} \frac{\mu_1 - \mu_2 \cos\theta}{\mu_2 \sin\theta} \tag{3}$$

$$\phi_2 = (\frac{\pi}{2} - \phi_1) + \mathrm{TH1} \tag{4}$$

$$\phi_3 = \phi_2 - \mathrm{PHR} \tag{5}$$

The refractive indicies are given by the A-H equation:



Figure 7.1 Geometry used in the standing analysis, formulated from the Appleton-Hartree equation.

$$\mu_1 = \frac{1 - 2X(1-X)}{2(1-X) - (y_{T_1})^2 + [y_{T_1}{}^2 + 4\, y_{L_1}{}^2(1-X)^2]^{1/2}} \tag{6}$$

$$\mu_2 = \frac{1 - 2X(1-X)}{2(1-X) - (y_{T_2})^2 + [y_{T_2}{}^2 + 4\, y_{L_2}{}^2(1-X)^2]^{1/2}} \tag{7}$$

where

$$y_{L_1} = y\ \mathrm{CO1}$$

$$y_{L_2} = y\ \mathrm{CO2}$$

$$y_{T_1} = y\ \mathrm{STI}$$

$$y_{T_2} = y\ \mathrm{ST2}$$

CO1, STI = cosine, and sine of the angle between $\vec{B}$ and $\vec{r}_1$

CO2, ST2 = cosine, and sine of the angle between $\vec{B}$ and $\vec{r}_2$.

Electron concentrations are deduced from standing wave pattern by an iterative process which calculates the distance between the fades for a given value of X, and keeps changing X until the calculated distance agrees with the observed value.

7.2 Rocket Trajectory

In order to eliminate the necessity of entering rocket coordinates for a large number of heights, the rocket trajectory is calculated internally as follows.

The acceleration, g, due to gravity, is assumed to decrease linearly over a limited height range so that

$$g = g_0 - g_1(h - h_0)$$

where $g = g_0$ at $h = h_0$. This is a good approximation over the standing-wave region.

The equation of vertical motion is then

$$\frac{d^2h}{dt^2} = - g.$$

Solving this and using the boundary conditions that

$$v_h = v_c \text{ and } h = h_1 \text{ at } t = t_0,$$

we get for the time variation of the height of the rocket:

$$h = (h_1 - h_0 - \frac{g_0}{g_1})\ g_1^{1/2} \sin h\ \{g_1^{1/2}(t - t_0)\} + v_c \cos h\ \{g_1^{1/2}(t - t_0)\}.$$

Variation of horizontal distance with time is given by

$$(x - x_1) = (t - t_0)v_x$$

where $x = x_1$ at $t = t_0$ and v_x is the horizontal velocity (assumed constant). The coordinates h_1, x_1, v_c, v_x at time t_0 (time after launch) are input parameters to the program.

7.3 The Computer Program

The computer program is written in FORTRAN IV and will be discussed with the aid of the flow diagram of Figure 7.2. The names of variables and their definitions are listed in Table 18. Parameters are entered by statements (19-1) to (19-22), next, the trajectory parameters are calculated (19-23) to (19-31) and then part of the magnetoionic parameters are determined (19-32) to (19-46). The input data consisting of propagation angles for the direct and reflected rays and the elapsed time between successive fades are entered at this point (19-47) to (19-53). For each item of the standing-wave data a corresponding electron density is determined in the following manner: First the position and velocity of the rocket is calculated (20-5) to (20-12), next, the observed distance between the fades is determined (20-13) to (20-15), the propagation angles at the rocket are interpolated from the input data (20-16) to (20-26), and then the magnetoionic parameters Y_L and Y_T for the A.H. equation are



Figure 7.2 Simplified flow diagram for the standing-wave analysis computer program.

TABLE 18

Names of Variables

AZ	Azimuth angle of the propagation plane (an input parameter)
AZI	Geomagnetic azimuth angle of the propagation plane
B1,B2,B3	Direction cosines of $\vec{B}$
BD,BN,BW	Vertical, northward, and westward components of $\vec{B}$ (input parameters)
BT	Magnetude of $\vec{B}$ (an input parameter)
CLI	cos(TH1)
CL2	cos(TH2
CN1	sin(TH1)
CN2	sin(TH2)
CO1	Cosine of the angle between the direct ray and $\vec{B}$
CO2	Cosine of the angle between the reflected ray and $\vec{B}$
CON	$\pi/180$
DCA	Calculated distance between fades, measured along y-axis
DCA	Calculated distance between fades, measured along the rocket trajectory
DIP	Magnetic dip angle
DM	Distance between fades if both rays were propagating in vacuum
DOB	Observed distance between fades
DX	Increment for changing X
EOM	Charge to mass ratio for electron
ED	Electron density (cm^{-3})
F	Frequency of the propagating wave
FH	
FR	Cosine of the angle between the direct and reflected rays
GO	Acceleration due to gravity, at height HO
GI	Rate of decrease of gravitational acceleration per meter elevation
HO	See GO
HI	Elevation of the rocket at time TO
HD(I)	See THD(I)
HU(I)	See THU(I)
N1	Number of points where propagation angles for the direct ray is given (an input parameter)
N2	Number of points where propagation angles for the reflected ray is given (an input parameter)
N3	Number of points where standing wave data is given (an input parameter)
PI	π
PHR	Rocket propagation angle measured from the horizontal axis
ST1	Sine of the angle between the direct ray and $\vec{B}$
ST2	Sine of the angle between the reflected ray and $\vec{B}$
SW(I)	Time lapse between two fades, occuring at mean time T(I) (input data)
T(I)	See SW(I)
TO	Initial time (seconds after launch) for calculating the rocket trajectory
TH1	Interpolated value of the direct ray propagation angle at the rocket
TH2	Interpolated value of the reflected ray propagation angle at the rocket
THD(I)	Propagation angle of the reflected ray at a height HD(I) (input data)
THU(I)	Propagation angle of the direct ray at a height HU(I)
UM1	Square of the refractive index of direct ray
UM2	Square of the refractive index of reflected ray

TABLE 18 (Continued)

VC	Vertical velocity of the rocket at time TO (an input parameter)
VX	Horizontal velocity of the rocket at time TO
WH	Gyrofrequency (rad/sec)
WL	Wave length in vacuum
X	$\omega_n{}^2/\omega^2 = Ne^2/\varepsilon_0 m\omega^2$
X1	Horizontal coordinate of the rocket at time TO (an input parameter)
XN	Ratio of X to electron density
Y	Ratio of gyrofrequency to propagation frequency
YL1,YL2	Longitudinal component of Y for the direct and reflected rays
YT1,YT2	Transverse component of Y for the direct and reflected rays

TABLE 19

```
C     STANDING WAVE ANALYSIS, A-H
        DIMENSION HU(50),THU(50),HD(50),THD(50),T(200),SW(200)
C     ENTER CONSTANTS
        PI = 3.1415927
        CON = PI/180.
        EOM = 1.758796E11
C     ENTER PARAMETERS
        F = 3.385E06
        BD = 4.894E-05
        BN = 1.723E-05
        BW = 3.549E-06
        BT = 5.201E-05
        AZ = 94.67
        N1 = 12
        N2 = 10
        N3 = 105
        TO = 88.33
        H1=106742.
        X1=27726.
        VO=1139.3
        VX=367.94
        HO=152400.
C     CALCULATE TRAJECTORY PARAMETERS
        GO=9.351
        G1=2.76E-6
        C1=GO/G1+HO
        GS=SQRT(G1)
        C2=((H1-C1)*GS+VO)/2.
        C3=((C1-H1)*GS+VO)/2.
        C4=(H1-C1-VO/GS)/2.
        C5=C2/GS
C     CALCULATE PRELIMINARY VARIABLES
        WL=2.997925E8/F
        W=2.*PI*F
        XN = 3.182407E3/W/W
        DIP = ATAN (BD/SQRT(BN*BN+BW*BW))
        DEC = ATAN(BW/BN)
        AZ = CON *AZ
        AZI = AZ+DEC
        WH = -BT*EOM
        Y = WH/W
        C=COS(DIP)
C     DIRECTION COSINES OF FIELD
        B1 = C*COS(AZI)
        B2 = C*SIN(AZI)
        B3 = -SIN(DIP)
C     ENTER DATA
        READ (5,3)  (HU(I),THU(I), I=1,N1)
  3     FORMAT(10F7.0)
        READ (5,5) (HD(I), THD(I), I=1,N2)
  5     FORMAT(10F7.0)
        READ (5,7) (T(I), SW(I),I=1,N3)
  7     FORMAT(10F7.0)
C     CONVERT KM TO M, DEG TO RAD.
        DO 4 I=1,N1,1
        HU(I)=HU(I)*1.E3
  4     THU(I)=THU(I)*CON
        DO 6 I=1,N2,1
        HD(I)=HD(I)*1.E3
```

TABLE 20

```
6     THD(I)=THD(I)*CON
C   PROCESS ALL DATA, N3 CASES.
      DO 8 M=1,N3,1
      TD=T(M)
C   COMPUTE TRAJECTORY
47    EX=GS*(T(M)-TO)
      E1=EXP(EX)
      E2=EXP(-EX)
      ALT=C4*E2+C5*E1+C1
      VH=C2*E1+C3*E2
      V=SQRT(VH*VH+VX*VX)
      PHR=ATAN(VH/VX)
C   DISTANCE OBSERVED BETWEEN FADES
      SPF=-SW(M)
      DOB=V*SPF
C   INTERPOLATE ARRIVAL ANGLES
      I=1
12    IF(HU(I)-ALT)9,10,10
9     I=I+1
      GO TO 12
10    J=1
14    IF(HD(J)-ALT)11,13,13
11    J=J+1
      GO TO 14
13    TH1=THU(I-1)+(ALT-HU(I-1))*(THU(I)-THU(I-1))/(HU(I)-HU(I-1))
      TH2=THD(J-1)+(ALT-HD(J-1))*(THD(J)-THD(J-1))/(HD(J)-HD(J-1))
C   CALCULATE PROPAGATION ANGLES
      CL1=COS(TH1)
      CN1=SQRT(1.-CL1*CL1)
      IF(TH1)15,16,16
15    CN1=-CN1
16    CL2=COS(TH2)
      CN2=SQRT(1.-CL2*CL2)
      IF(TH2)17,18,18
17    CN2=-CN2
18    CO1 = B1*CL1+B3*CN1
      CO2 = B1*CL2+B3*CN2
      YL1 = Y*CO1
      YL2 = Y*CO2
      SI1=SQRT(1.-CO1*CO1)
      SI2=SQRT(1.-CO2*CO2)
      YT1 = Y*SI1
      YT2 = Y*SI2
      YT12=YT1*YT1
      YT22=YT2*YT2
      FR=CN1*CN2+CL1*CL2
      TL=SQRT(1.-FR*FR)
C   TEST OF VACUUM FADING DISTANCE
      DM=WL/SQRT(2.-2.*FR)
      PH=(TH1+TH2+PI)/2.-PHR
      DM=DM/COS(PH)
      IF(DM)52,53,53
52    DM=-DM
53    IF(DM-DOB)70,70,66
66    FD=-1.
      GO TO 8
C   REFRACTIVE INDICES FROM A-H EQS.
70    X=.5
      DX=.11
      I=1
```

calculated (Y_{L_1}, Y_{T_1} for the direct ray and Y_{L_2}, Y_{T_2} for the reflected ray) (20-36) to (20-45). At this point the vacuum fading distance DM (the distance that would be observed between fades if the rays were propagating in vacuum) is calculated and compared with the observed distance DOB (20-48) to (20-56), if the latter is equal to or smaller than the former no meaningful electron density can be calculated and the program goes to the next data. If the observed fading distance is larger than the vacuum fading distance the corresponding electron density is determined by an iterative process which varies X until the observed and calculated distance between the fades agree. The iteration loop starts with calculating the refractive indicies for the direct and reflected rays (21-3) to (21-12), next, the distance between the fades DCA is calculated (21-13) to (21-20), and then the observed and calculated distances are compared, if they differ by more than 0.1%, X is incremented (21-21) to (21, 42) and DCA is recalculated, the iteration continues until the observed and calculated distances agree within 0.1 %; then the electron density ED is determined from the final value of X (21-43) and the result printed (21-44). Now the program returns to (20-3) to process the next standing-wave data.

TABLE 21

```
      J=0
32    X1=(1.-X)*2.
      D1=X1-YT12+SQRT(YT12*YT12+YL1*YL1*X1*X1)
      D2=X1-YT22+SQRT(YT22*YT22+YL2*YL2*X1*X1)
      UM1=1.-X*X1/D1
      UM2=1.-X*X1/D2
      IF(UM1)33,63,63
63    IF(UM2)33,64,64
64    ER=2.*FR*SQRT(UM1*UM2)
      DCA=WL/SQRT(UM1+UM2-ER)
      UM1S=SQRT(UM1)
      UM2S=SQRT(UM2)
      PH=ATAN((UM1S-UM2S*FR)/UM2S/TL)
      IF(PH)19,20,20
19    PH=PH+PI
20    PH=TH1-PH+PI/2.
      PH=PH-PHR
      DCA=DCA/COS(PH)
      IF(DCA)60,61,61
60    DCA=-DCA
C     MATCH CALCULATED AND OBSERVED DISTANCES.
61    P=(DCA-DOB)/DOB
      IF(I-1)21,21,22
22    DX=DX/2.
21    IF(P)23,24,24
23    IF(P+.001)25,26,26
25    X=X+DX
      IF(J-1)27,28,29
28    I=2
27    J=2
29    IF(X)30,31,31
31    IF(X-1.)32,33,33
24    IF(P-.001)26,26,35
35    X=X-DX
      IF(J-1)34,29,36
36    I=2
34    J=1
      GO TO 29
30    DX=DX/2.
      GO TO 25
33    DX=DX/2.
      GO TO 35
26    ED=X/XN
8     WRITE(6,67)TD,ALT,ED,DOB,DCA,UM1S,UM2S,PHR
67    FORMAT (1P8E15.5)
      END
```

8. STANDING-WAVE ANALYSIS, FORMULATED FROM THE BOOKER QUARTIC EQUATION

An alternate, but physically equivalent, method of analyzing the standing-wave data makes use of the Booker quartic equation. Roots of the Booker quartic equation (Booker, 1938; Budden, 1961; Kelso, 1964) conveniently relate refractive indices and arrival angles of waves propagating obliquely in a doubly refracting ionosphere.

Assuming horizontal stratification of the ionosphere, we may write Snell's law for the direct and reflected rays respectively as follows.

$$\mu_i \sin \theta_i = \mu_{io} \sin \theta_{io} \tag{8.1}$$

$$\mu_r \sin \theta_r = \mu_{ro} \sin \theta_{ro} \tag{8.2}$$

or in the computer notation, defined in Table 22,

$$\text{UI} \sin \text{THI} = \sin \text{THIC} \equiv \text{SI}, \tag{8.3}$$

$$\text{and UR} \sin \text{THR} = \sin \text{THRC} \equiv \text{SR}. \tag{8.4}$$

For a trial value of X (if sufficiently close to the correct value), two roots of the Booker equation are pure real numbers, corresponding to the upgoing and downgoing rays, from which refractive indices are obtained by the equations,

$$\text{UI} = \sqrt{(\text{QIR})^2 + (\text{SI})^2}\ , \tag{8.5}$$

$$\text{and UR} = \sqrt{(\text{QRR})^2 + (\text{SR})^2}\ . \tag{8.6}$$

Then, since THIC and THRC are known, in first approximation, from an initial set of ray tracings,

$$\text{THI} = \arcsin\ (\text{SI}/\text{UI}), \tag{8.7}$$

$$\text{and THR} = \arcsin\ (\text{SR}/\text{UR}). \tag{8.8}$$

Wavelengths in the directions of the direct and reflected rays are also obtained from the refractive indices and the equations,

$$\text{WLI} = \text{C}/(\text{F} * \text{UI}), \tag{8.9}$$

TABLE 22

Names of Variables

Column No. 1	2	3	
AL	-	α_i	Coefficient of Booker quartic equation for direct wave
AL	-	α_r	Coefficient of Booker quartic equation for reflected wave
BEI	-	β_i	Coefficient of Booker quartic equation for direct wave
BER	-	β_r	Coefficient of Booker quartic equation for reflected wave
BT	B	B	The total flux density of the geomagnetic field (tesla)
BX	BX	B_x	Geomagnetic field component parallel to trajectory plane (tesla)
BY	BY	B_y	Geomagnetic field component perpendicular to trajectory plane (tesla)
BZ	BA	B_z	Upward component of geomagnetic field (tesla)
C	-	c	Speed of light in vacuum
CI	-	-	cos (THIC)
CR	-	-	cos (THRC)
D	N	N	Electron density (m^{-3})
DEI	-	δ_i	Coefficient of Booker quartic equation for direct wave
DER	-	δ_r	Coefficient of Booker quartic equation for reflected wave
E	-	e	Charge of an electron, a positive number (coulomb)
EPI	-	ε_i	Coefficient of Booker quartic equation for direct wave
EPR	-	ε_r	Coefficient of Booker quartic equation for reflected wave
F	F	f	Frequency of propagating wave (Hz)
GAI	-	γ_i	Coefficient of Booker equation for direct wave
GAR	-	γ_r	Coefficient of Booker quartic equation for reflected wave
H	H	z	Altitude above the earth (km)
N	N	n	The number of iterations performed
P	-	ε_o	Permittivity of free space (farad/meter)
PH	-	ϕ	Elevation angle from the transmitter to the rocket (radians)
PHD	EL	ϕ	Elevation angle from the transmitter to the rocket (degrees)
QII	-	q_i	A root of the Booker equation, imaginary part, direct wave
QIR	QD	q_i	A root of the Booker equation, real part, direct wave
QRI	-	q_r	A root of the Booker equation, imaginary part, reflected wave
QRR	QR	q_r	A root of the Booker equation, real part, reflected wave
RL	RL	ℓ	Calculated distance along rocket path between minima (meters)
RLO	RLO	-	Observed distance along rocket path between minima (meters)
SI	-	-	sin (THIC)
SR	-	-	sin (THRC)
THI	-	θ_i	Arrival angle of the direct wave (radians)
THIC	-	θ_{io}	Incident angle of the direct wave (radians)
THICD	THD	θ_{io}	Incident angle of the direct wave (degrees)
THID	PSD	θ_i	Arrival angle of the direct wave (degrees)
THR	-	θ_r	Arrival angle of the reflected wave (radians)
THRC	-	θ_{ro}	Incident angle of the reflected wave (radians)
THRCD	THR	θ_{ro}	Incident angle of the reflected wave (degrees)
THRD	PSR	θ_r	Arrival angle of the reflected wave (degrees)
UI	MUD	μ_i	Refractive index of the direct wave

TABLE 22 (Continued)

UR	MUR	μ_r	Refractive index of the reflected wave
V	-	-	$e/2\pi fm$
VC	-	c	Speed of light in vacuum (meter/second)
W	-	m	Mass of the electron (kilogram)
WLI	WLD	λ_i	Wavelength of the direct wave (meters)
WLR	WLR	λ_r	Wavelength of the reflected wave (meters)
X	X	X	Square of the ratio of plasma frequency to wave frequency
Y	-	Y	Ratio of gyrofrequency to wave frequency

Column No. 1 lists the names of variables used in the source program (Tables 23, 24, and 25, Column No. 2 lists corresponding names used to identify output variables, and Column No. 3 lists conventional mathematical symbols corresponding to the variables of Columns 1 and 2. Definitions of each of the variables follow to the right.

$$\text{and WLR} = C/(F * UR). \qquad (8.10)$$

Constant phase fronts of the direct and reflected waves are represented in Figure 8.1 as they intersect in the vicinity of the rocket and establish surfaces of minimum field strength. The geometry of Figure 8.1 dictates the following sequence of equations:

$$SH = \sin\ (THR + THI) \qquad (8.11)$$

$$CH = \cos\ (THR + THI) \qquad (8.12)$$

$$SOL = \frac{WLI}{SH} \qquad (8.13)$$

$$SSH = \frac{WLR}{SH} \qquad (8.14)$$

$$DIAG = \{SOL^2 + SSH^2 + 2\ SOL * SSH * CH\}^{1/2} \qquad (8.15)$$

$$\cos TA \equiv CTA = \frac{SOL^2 + DIAG^2 - SSH^2}{2 * DIAG * SOL} \qquad (8.16)$$

$$\sin TA \equiv STA = \sqrt{1 - CTA ** 2} \qquad (8.17)$$

$$TA = \arctan \frac{STA}{CTA} \qquad (8.18)$$

$$S = SOL * STA \qquad (8.19)$$

$$SRL = \sin(PH + TA - THR) \qquad (8.20)$$

$$RL = \frac{S}{SRL} \qquad (8.21)$$

The calculated value of RL, the distance along the rocket path between two adjacent minima of signal strength, is of principal interest. A correct value of X will bring RL into agreement with the observed distance between minima, RLO. The trial value of X is changed until

$$(RLO - RL)/RLO \leq 0.001, \qquad (8.22)$$

An iteration equation, for changing the value of X until (8.22) is satisfied, is derived as follows. For the ordinary mode of propagation,



Figure 8.1 Intersection of direct and reflected waves.

$$\lambda = \frac{c}{f}\frac{1}{\mu} = \frac{c}{f}\left[1 - \frac{X}{1 + Y_L}\right]^{1/2} . \tag{8.23}$$

Thus, changes in λ and X are related by the equation

$$\Delta\lambda = \frac{c}{2f} \frac{\Delta X}{(1 + Y_L)\left(1 - \frac{X}{1 + Y_L}\right)^{3/2}} \tag{8.24}$$

From general studies of standing waves in the E region, we have observed that

$$RL \approx 1/2\ \lambda \quad \text{or} \quad \Delta(RL) \approx 1/2\ \Delta\lambda \tag{8.25}$$

where λ is the wavelength of the direct or upgoing wave. According to this approximation, the change in X required to produce a desired change in RL is given by the equation

$$\Delta X = \frac{4f(1+Y_L)}{c}\left(1 - \frac{X}{1+Y_L}\right)^{3/2} (RLO - RL) \tag{8.26}$$

The FORTRAN equivalent of (8.26), which iterates X until (8.22) is satisfied, is

```
X = 2. * F/VC * (1.+YL) * (1. - X/(1.+YL)) ** 1.5 * (RLO-RL) + X   (8.27)
```

The convergence of RL to RLO was found to improve by a change of the numerical coefficient 4 of (8.26) to 2 of (8.27). This reduction in the magnitude of X tends to limit "over correction" of RL. Convergence is usually attained in three to five iterations.

However, difficulties are encountered when X is made to advance into the forbidden region, X > 1, by the iteration equation, (8.27). Under these circumstances, X if forced to remain less than one by repeated subtraction of the number 0.01, but convergence is still not attained in some cases. If convergence is not achieved, RL is calculated and plotted as a function of X. The value of X corresponding to RLO is then read from the graph.

Values of electron density, obtained by the procedure described above, are used for a second ray tracing to calculate more accurate angles of incidence for the direct and reflected rays. These, in turn, are used in (8.5) through (8.27) to obtain more correct values of X and of electron density.

The changes of incidence and arrival angles resulting from a third ray tracing are generally less than ±1°, and have a relatively insignificant influence on the calculated values of electron density. Numerical calculations of variations are illustrated graphically in Figure 8.2. From these calculations we conclude that:

1. A 1° positive change in the value of THI results in an 0.5% decrease in electron density, at most.
2. A 1° positive change in the value of THR results in a 1.5% decrease in electron density, at most.
3. A 1% positive change in the value of the quantity, RLO, results is a 3% increase in electron density.

It is evident that the limiting factor is the accuracy of the observed distances between fades, RLO, and not the determination of arrival angles. A third ray tracing cannot improve the accuracy of electron densities.

The FORTRAN IV program which implements (8.5) through (8.27) is listed in Tables 23, 24, and 25. Parameters for a given rocket shot are entered by Statements (23-10) to (23-18). Altitude, incident angles of the direct and reflected rays, and the observed spacing of fades are entered for each data point by (23-19). Coefficients of the Booker equation are calculated by (23-44) through (24-22). Equations (8.5) through (8.21) are represented by Statements (24-24) through (24-46), and the iteration and output statements follow from (24-47) through (25-11).



Figure 8.2 Variation of electron density with parameters of the analysis.

TABLE 23

```
C   STANDING WAVE ANALYSIS BY BOOKER QUARTIC
      DIMENSION R(8)
C   READ CONSTANTS
      E=1.602E-19
      P=8.8542E-12
      W=9.1091E-31
      VC=2.9979E8
      PI=3.1415927
      RD = 1.74533E-02
C   READ PARAMETERS
      PHD = 7.7599E01
      AZD = 1.5895E02
      BZ = 1.3369E-05
      BS =-2.2563E-05
      BE =-2.5310E-06
      BT = 2.6348E-05
      YL =-6.9662E-02
      F = 3.3850E06
 21   READ(5,42) H,THRCD,THICD,RLO
 42   FORMAT (4F10.2)
C   CONVERT DEGREES TO RADIANS
      PH=PHD*RD
      THIC=THICD*RD
      THRC=THRCD*RD
      AZ = AZD*RD
C   COMPUTE INTERMEDIATE VARIABLES
      BX = BS*SIN(AZ)+BE*COS(AZ)
      BY = -BS*COS(AZ)+BE*SIN(AZ)
      V= 1.758796E11/2./PI/F
      YX = V*BX
      YY = V*BY
      YZ = V*BZ
      Y=V*BT
      AK=E*E/((W*F**2)*(P*4.*PI**2))
      X=.7
      N=0
 35   N=N+1
      IF(N-35) 36,38,38
 36   CONTINUE
      SI = SIN(THIC)
      SR = SIN(THRC)
      CI = COS(THIC)
      CR = COS(THRC)
C   CALCULATE COEFFICIENTS OF QUARTIC EQUATION
      AL=(1.-Y**2)-X*(1.-YZ**2)
      BEI=SI*X*YY*YZ*2.
      BER=SR*X*YY*YZ*2.
      GAI=(CI**2*Y**2-(1.-X)*(CI**2-X))*2.-X*(YX**2+CI**2*YY**2+(1.+CI
     1**2)*YZ**2)
      GAR=(CR**2*Y**2-(1.-X)*(CR**2-X))*2.-X*(YX**2+CR**2*YY**2
     1+(1.+CR**2)*YZ**2)
      DEI=-CI**2*SI*YZ*YY*2.
      DER=-CR**2*SR*YZ*YY*2.
      EPI=(1.-X)*((CI**2-X)**2-YY**2*CI**4)-CI**2*(CI**2
     1-X)*(YX**2+YZ**2)
      EPR=(1.-X)*((CR**2-X)**2-YY**2*CR**4)-CR**2*(CR**2-X)*(YX**2
     1+YZ**2)
C   FIND ROOTS OF QUARTIC EQUATION FOR DIRECT WAVE
      CALL QUAR(AL,BEI,GAI,DEI,EPI,R)
      IF(R(8)) 5,3,5
```

TABLE 24

```
3       QD=R(7)
        GO TO  61
5       IF (R(6))  6,14,6
14      QD=R(5)
        GO TO  61
6       IF(R(4))   7,2,7
2       QD=R(3)
        GO TO  61
7       QD=R(1)
61      CONTINUE
C  FIND ROOTS OF QUARTIC EQUATION FOR REFLECTED WAVE
        CALL QUAR(AL,BER,GAR,DER,EPR,R)
        IF(R(2))17,16,17
16      QR=R(1)
        GO TO 91
17      IF(R(4)) 19,18,19
18      QR=R(3)
        GO TO 91
19      IF(R(6)) 22,20,22
20      QR=R(5)
        GO TO 91
22      QR=R(7)
91      CONTINUE
C  CALCULATE DIRECT WAVE INDEX, ARRIVAL ANGLE, AND WAVELENGTH
8       UI=SQRT(QD**2+SI*SI)
        THI=ARSIN(SI/UI)
        THID=THI/RD
        WLI=VC/F/UI
C  CALCULATE REFLECTED WAVE INDEX, ARRIVAL ANGLE, AND WAVELENGTH
        UR=SQRT(QR**2+SR*SR)
        THR=ARSIN(SR/UR)
        THRD=THR/RD
        WLR=VC/F/UR
C  COMPUTE SUPERPOSITION OF DIRECT AND REFLECTED WAVES
4       SH = SIN(THR+THI)
        CH = COS(THR+THI)
        SOL=WLI/SH
        SSH=WLR/SH
        DIAG2=SOL**2+SSH**2+CH*SOL*SSH*2.
        DIAG=SQRT(DIAG2)
        CTA = (SOL**2+DIAG2-SSH**2)/(DIAG*SOL*2.)
        STA=SQRT(1.-CTA**2)
        TA =ATAN(STA/CTA)
        S=SOL*STA
        SRL = SIN(PH+TA-THR)
        RL=S/SRL
        IF(ABS((RLO-RL)/RLO)-0.001) 38,38,39
38      CONTINUE
C  PRINT VALUES OF IMPORTANT QUANTITIES
        WRITE(6,71)F,PHD,D,X,H
        WRITE(6,72)BT,BX,BY,BZ
        WRITE(6,73)THICD,THID,THRCD,THRD
        WRITE(6,74)QD,QR,UI,UR
        WRITE(6,75)WLI,WLR,S,RL
        WRITE(6,76)SOL,SSH,N,RLO
        GO TO 21
39      X = 2.*F/VC*(1.+YL)*(1.-X/(1.+YL))**1.5*(RLO-RL)+X
25      IF (X-0.98)  70,70,80
80      X = X-0.01
        GO TO 25
```

TABLE 25

```
 70    CONTINUE
       D=X/AK
 71    FORMAT(6H   F =1PE10.3,5X,5H EL =0PF5.1,10X,5H  N =1PE10.3,5X,
      15H  X =0PF6.3,6X,4HH = F6.2)
 72    FORMAT(6H   B =1PE14.7,1X,5H BX =1PE14.7,1X,5H BY =1PE14.7,1X,
      15H BZ =1PE14.7)
 73    FORMAT(6H THD =F6.3,9X,5HPSD =F6.3,9X,5HTHR =F5.1,10X,
      15HPSR =F5.1)
 74    FORMAT(6H  QD =F6.4,9X,5H QR =F7.4,8X,5HMUD =F6.4,9X,5HMUR =F6.4)
 75    FORMAT(6H WLD =F6.1,9X,5HWLR =F6.1,9X,5H  S =F6.1,9X,5H RL =F6.2)
 76    FORMAT(6H SOL =F6.1,9X,5HSSH =F6.1,10X,4H N =I2,13X,5HRLO =F6.2/)
       GO TO 35
       END
```

REFERENCES

Booker, H. G. (1938), Propagation of wave-packets incident obliquely upon a stratified doubly refracting ionosphere, Phil. Trans. A 237, 411-450.

Bowhill, S. A. (1965), IQSY rocket studies of the C, D, and E regions of the ionosphere, NASA University Program Review Conference, NASA Special Publication SP-85 (U. S. Govt. Printing Office, Washington, D. C.), 269-275.

Budden, K. G. (1961), *Radio Waves in the Ionosphere*, Cambridge University Press, Great Britain.

Cain, J. C., S. J. Hendricks, R. A. Langel and W. V. Hudson (1967), A proposed model for the international geomagnetic reference field-1965, J. Geomagn. Geoelect. 19, 335-355.

Cain, S. H., J. C. Cain, W. E. Daniels and D. C. Jensen (1968), Computation of the main geomagnetic field from spherical harmonic expansions, NSSDC 68-11, National Space Science Data Center, Greenbelt, Maryland.

Gill, S. (1951), A process for the step-by-step integration of differential equations in an automatic digital computing machine, Proc. Camb. Phil. Soc. 47, 96-108.

Hara, E. H. (1963), Approximations to the semiconductor integrals $C_p(x)$ and $D_p(x)$ for use with the generalized Appleton-Hartree magnetoionic formulas, J. Geophys. Res. 68, 4388-4389.

Haselgrove, J. (1954), Ray theory and a new method of ray tracing, *The Physics of the Ionosphere*, The Physics Society, London, 355-364.

Kelso, J. M. (1964), *Radio Ray Propagation in the Ionosphere*, McGraw-Hill Book Co., New York.

Knoebel, H. W. and D. O. Skaperdas (1966), Rocket measurements of Faraday rotation and differential absorption, Rev. Sci. Instr. 37, 1395-1400.

Mechtly, E. A., S. A. Bowhill, L. G. Smith and H. W. Knoebel (1967), Lower ionosphere electron concentration and collision frequency from rocket measurements of Faraday rotation, Differential absorption, and probe current, J. Geophys. Res. 72, 5239-5245.

Mechtly, E. A. (1969), The international system of units, physical constants, and conversion factors, NASA SP 7012, revised, U. S. Govt. Printing Office.

Monro, P. E., E. A. Mechtly and R. S. Sastry (1968), Analysis of ordinary mode standing waves observed by rockets in the E-region, J. Geophys. Res. 73, 4291-4297.

Phelps, A. V. and J. L. Pack (1959), Electron collision frequencies in nitrogen and in the lower ionosphere, Phys. Rev. Letters 3, 340-342.

Sen, H. K. and A. A. Wyller (1960), On the generalization of the Appleton-Hartree magnetoionic formulas, J. Geophys. Res. 65, 3931-3950.

Taylor, B. N., W. H. Parker and D. N. Langenberg (1969), Determination of e/h, using macroscopic quantum phase coherence in superconductors: implications for quantum electrodynamics and the fundamental physical constants, Rev. of Modern Physics 41, 375-496.